COMPASS for Pressure (or Flow)
Alert if Piston Gauge’s piston-cylinder temperature is outside of limits

Create a test macro to set piston-cylinder temperature limits and specify what to do if conditions are outside the limits



'**************************************************
'Test Macro do not have a return value.
'Manipulate the test or device collection as desired. 
'iT  The current temperature point in the test
'iL  The current line pressure point in the test.
'iC  The current pressure cycle in the test.
'IP  The current pressure point in the test.
'cTest   The test class .
'cConfig  Configuration class that holds all active devices.
'**************************************************

Function PC_TempLimit_Monitor(iT, iL, iC, iP, cTest, cConfig)

'enter e-mail address if using the e-mail notification feature
[bookmark: _GoBack]'The method used relies on MS Outlook(R) being open when running the test
'comment out the email lines if not using this feature
emailto = "noreply@flukecal.com"

Delay = 900 '(sec), time to wait between notifications

PC_TempLL = 15
PC_TempHL = 25
PC_TempLtxt = "C"	'Set to C or F
PC_TempLunit = 0	'Set to corresponding integer; C=0, F=1, K=2, R=3
'	cDebug.LogStatus "PC Temp Low Limit: " & PC_TempLL & " " & PC_TempLtxt & " (" & PC_TempLunit & ")"
'	cDebug.LogStatus "PC Temp High Limit: " & PC_TempHL & " " & PC_TempLtxt & " (" & PC_TempLunit & ")"
PC_Temp = cCOMPASS.cConfig.RefPrs(1).RangeRaw1.FinalOutput
PC_TempU = cCOMPASS.cConfig.RefPrs(1).RangeRaw1.UnitFinal
PC_TempUtxt = cCOMPASS.cConfig.RefPrs(1).RangeRaw1.UnitFinalText	'bug, this results in mV when units are F
PC_TempUtxt = PC_TempLtxt
'	cDebug.LogStatus "PC_Temp = " & PC_Temp & " " & PC_TempUtxt & " (" & PC_TempU & ")"
PC_TempLL = cCOMPASS.UnitConversion (CDbl(PC_TempLL),CInt(PC_TempU),CInt(PC_TempLunit),1)
'	cDebug.LogStatus "PC_TempLL = " & PC_TempLL & " " & PC_TempUtxt & " (" & PC_TempU & ")"
PC_TempHL = cCOMPASS.UnitConversion (CDbl(PC_TempHL),CInt(PC_TempU),CInt(PC_TempLunit),1)
'	cDebug.LogStatus "PC_TempHL = " & PC_TempHL & " " & PC_TempUtxt & " (" & PC_TempU & ")"
If PC_Temp < PC_TempLL  Then
'	cDebug.LogStatus "PC Temp Low at " & PC_Temp & " " & PC_TempUtxt 
	msgPCT = "Piston-Cylinder Temperature is too low at " & PC_Temp & " " & PC_TempUtxt & ". "
	msgPCT = msgPCT & "PC_TempLL = " & PC_TempLL & " " & PC_TempUtxt
	PCTalert msgPCT,emailto,delay
ElseIf PC_Temp > PC_TempHL Then 
'	cDebug.LogStatus "PC Temp High at " & PC_Temp & " " & PC_TempUtxt
	msgPCT = "Piston-Cylinder Temperature is too high at " & PC_Temp & " " & PC_TempUtxt & ". "
	msgPCT = msgPCT & "PC_TempHL = " & PC_TempHL & " " & PC_TempUtxt
	PCTalert msgPCT,emailto,delay
End If

End Function

Function PCTalert (msgPCT,toEmail,delay)
	msgPCT = msgPCT & ". Wait " & formatnumber(delay/60,1) & " minutes until checking again. "
	cCOMPASS.StatusDisplay msgPCT
	cDebug.LogStatus msgPCT
	msgbox msgPCT
	msgPCT = msgPCT & "Current date/time is  " & Date & ", " & time & "."
	subject = msgPCT
   	SendEmail toEmail, subject, msgPCT
   		cDebug.LogStatus "Sent email to: " & toEmail
	TimeDelayWstatus (delay)	
End Function

The above macro calls the Global macro TimeDelayWstatus. TimeDelayWStatus should be added as a Global macro
'****************************************
'Delay for the amount of time in seconds
'specified by tD. Use the COMPASS object
'to allow for asynchronous operation
'****************************************
Sub TimeDelayWstatus(tD)
	tStart = Timer'Timer function returns the number of seconds since 12:00 AM
	Do 			
		cCOMPASS.TimeDelay 1
		tDiff = Time_Difference(tStart)
		remainT = CInt(tD - tDiff)
		cCOMPASS.StatusDisplay "Delay remaining..." & remainT
		If cCOMPASS.SystemAbort Then Exit Sub
		If remainT <= 0 Then Exit Do
	Loop Until False
End Sub

And this macro calls the Global macro TimeDelay. It is already in COMPASS as a Global macro but the text is here if you need to add it.
'****************************************
'Delay for the amount of time in seconds
'specified by tD. Use the COMPASS object
'to allow for asynchronous operation
'****************************************
Sub TimeDelay(tD)
cCOMPASS.TimeDelay tD
 End Sub

Same with Global macro Time_Difference
'///////////////////////////////////////////////////////////////
'Return the time difference from the current
'time and the time value passed T:
'
'In the event the diff is <0 then add the
'number of seconds in a day because midnight
'has passed since the start time.
'
'NOTE
'This function as an effective time resoluteion
'of 10 ms.
'
'///////////////////////////////////////////////////////////////
Public Function Time_Difference(t) 
Dim T_Diff 

T_Diff = Timer - t  
If T_Diff < -0.1 Then
    T_Diff = T_Diff + 86400
ElseIf T_Diff < 0 Then
    T_Diff = 0
End If
 
Time_Difference = T_Diff

End Function

If using the email feature his macro calls this Global macro. Add it as a Global macro.

Function SendEmail(emailto,subject,body)

	Set MyApp = CreateObject("Outlook.Application")
	Set MyItem = MyApp.CreateItem(0)
	With MyItem
    	.To = emailto
    	.Subject = subject
    	.ReadReceiptRequested = False
    	.HTMLBody = body
	End With
MyItem.Send

End Function
